# Maintainer: Felix Yan <felixonmars@archlinux.org>

_hkgname=hls-brittany-plugin
pkgname=haskell-hls-brittany-plugin
pkgver=1.1.0.0
pkgrel=106
pkgdesc="Integration with the Brittany code formatter"
url="https://haskellwingman.dev"
license=("Apache")
arch=('loong64' 'x86_64')
depends=('ghc-libs' 'haskell-brittany' 'haskell-czipwith' 'haskell-extra' 'haskell-ghc'
         'haskell-ghc-exactprint' 'haskell-ghcide' 'haskell-hls-plugin-api' 'haskell-lens'
         'haskell-lsp-types')
makedepends=('ghc' 'uusi' 'haskell-hls-test-utils')
checkdepends=('git')
source=("https://hackage.haskell.org/packages/archive/$_hkgname/$pkgver/$_hkgname-$pkgver.tar.gz")
sha512sums=('b516e7ed4a5c4ba73a5b009f1618400d6bb7ad70a6a0868a3ef40d7fe655956dd65f4b915a5f5219101e3223fd71ff2ebef7fb907a20b03b6104cc33064c9315')

prepare() {
  cd $_hkgname-$pkgver
  gen-setup
}

build() {
  cd $_hkgname-$pkgver

  runhaskell Setup configure -O --enable-shared --enable-executable-dynamic --disable-library-vanilla \
    --prefix=/usr --docdir=/usr/share/doc/$pkgname --datasubdir=$pkgname --enable-tests \
    --dynlibdir=/usr/lib --libsubdir=\$compiler/site-local/\$pkgid --ghc-option=-fllvm \
    --ghc-option=-optl-Wl\,-z\,relro\,-z\,now \
    --ghc-option='-pie'

  runhaskell Setup build $MAKEFLAGS
  runhaskell Setup register --gen-script
  runhaskell Setup unregister --gen-script
  sed -i -r -e "s|ghc-pkg.*update[^ ]* |&'--force' |" register.sh
  sed -i -r -e "s|ghc-pkg.*unregister[^ ]* |&'--force' |" unregister.sh
}

check() {
  cd $_hkgname-$pkgver
  runhaskell Setup test --show-details=direct
}

package() {
  cd $_hkgname-$pkgver

  install -D -m744 register.sh "$pkgdir"/usr/share/haskell/register/$pkgname.sh
  install -D -m744 unregister.sh "$pkgdir"/usr/share/haskell/unregister/$pkgname.sh
  runhaskell Setup copy --destdir="$pkgdir"
  install -D -m644 LICENSE -t "$pkgdir"/usr/share/licenses/$pkgname/
  rm -f "$pkgdir"/usr/share/doc/$pkgname/LICENSE
}
